1.6.4 Numerical Differential Equations

`NDSolve[`*eqns*`, `*y*`, {`*x*`, `*xmin*`, `*xmax*`}]`	
	solve numerically for the function y, with the independent variable x in the range *xmin* to *xmax*
`NDSolve[`*eqns*`, {`y_1`, `y_2`, ... }, {`*x*`, `*xmin*`, `*xmax*`}]`	
	solve a system of equations for the y_i

Numerical solution of ordinary differential equations.

This generates a numerical solution to the equation $y'(x) = y(x)$ with $0 < x < 2$. The result is given in terms of an `InterpolatingFunction`.

```
In[1]:= NDSolve[{y'[x] == y[x], y[0] == 1}, y, {x, 0, 2}]
Out[1]= {{y -> InterpolatingFunction[{0., 2.}, <>]}}
```

Here is the value of $y(1.5)$.

```
In[2]:= y[1.5] /. %
Out[2]= {4.48191}
```

With an algebraic equation such as $x^2 + 3x + 1 = 0$, each solution for x is simply a single number. For a differential equation, however, the solution is a *function*, rather than a single number. For example, in the equation $y'(x) = y(x)$, you want to get an approximation to the function $y(x)$ as the independent variable x varies over some range.

Mathematica represents numerical approximations to functions as `InterpolatingFunction` objects. These objects are functions which, when applied to a particular x, return the approximate value of $y(x)$ at that point. The `InterpolatingFunction` effectively stores a table of values for $y(x_i)$, then interpolates this table to find an approximation to $y(x)$ at the particular x you request.

y`[`*x*`] /. `*solution*	use the list of rules for the function y to get values for *y*`[`*x*`]`
`InterpolatingFunction[`*data*`][`*x*`]`	
	evaluate an interpolated function at the point x
`Plot[Evaluate[`*y*`[`*x*`] /. `*solution*`]], {`*x*`, `*xmin*`, `*xmax*`}]`	
	plot the solution to a differential equation

Using results from `NDSolve`.

This solves a system of two coupled differential equations.

```
In[3]:= NDSolve[ {y'[x] == z[x], z'[x] == -y[x], y[0] == 0,
                               z[0] == 1}, {y, z}, {x, 0, Pi} ]
Out[3]= {{y -> InterpolatingFunction[{0., 3.14159}, <>],
   z -> InterpolatingFunction[{0., 3.14159}, <>]}}
```

Web sample page from The *Mathematica* Book, Second Edition, by Stephen Wolfram, published by Addison-Wesley Publishing Company (hardcover ISBN 0-201-51502-4; softcover ISBN 0-201-51507-5). To order *Mathematica* or this book contact Wolfram Research: info@wolfram.com; http://www.wolfram.com/; 1-800-441-6284.

Here is the value of z[2] found from the solution.

```
In[4]:= z[2] /. %
Out[4]= {-0.416167}
```

Here is a plot of the solution for y[x] found on line 1. Plot is discussed in Section 1.9.1.

```
In[5]:= Plot[Evaluate[y[x] /. %1], {x, 0, 2}]
```